

A Gentle Introduction on Lean

Moti Ben-Ari

`http://www.weizmann.ac.il/sci-tea/benari/`

Version 0.5

January 6, 2024

© Moti Ben-Ari, 2023

This work is licensed under a Creative Commons Attribution 4.0 International License. To view a copy of this license, visit <http://creativecommons.org/licenses/by/4.0/>.

Contents

1	Introduction	1
2	Commutativity of the greatest common denominator	4
3	Commutativity of minimum	6
4	Cancellation properties	8
5	There exists an infinite number of prime numbers	11
6	The square root of two is irrational	15
7	Propositional logic	19
8	Induction	25
A	Keyboard shortcuts	30
B	Tactics	31
C	Tips	33

1 Introduction

Lean 4 is a *proof assistant*. You enter your proof into Lean in a formal language and the system checks the correctness of the proof. It displays the current set of hypotheses and goals, and it is capable of performing many simple proofs automatically. This tutorial is, as its name states, a gentle introduction to Lean intended for students and others who have no previous experience with a proof assistant.

Constructs of Lean are explained and demonstrated within complete proofs of theorems of arithmetic and logic. The source code is heavily commented and additional explanations are given. As you gain more experience with Lean, the number of comments on a proof is reduced. Proof tactics are introduced one at a time as needed as are tips. Tables of keyboard shortcuts, tactics and tips are given in the appendices. The tips include information on definitions and notations used by Lean.

This tutorial limits itself to proving properties of integers and natural numbers, for example, theorems on greatest common denominators and prime numbers. Beyond that you need a basic knowledge of propositional and first-order logic as found in introductory textbooks.¹

The Lean source code of these proofs and the \LaTeX source of this document can be found at <https://github.com/motib/gentle-lean>.

Installation

To install Lean, follow the instructions for your operating system at

https://leanprover-community.github.io/get_started.html.

The Lean community uses *Visual Studio Code (VSC)* <https://code.visualstudio.com/>. It is a very versatile environment with lots of features, so be sure to study VSC tutorials before starting to work with Lean.

You *must* work within a project framework as described in

<https://leanprover-community.github.io/install/project.html>.

When you start Lean to work on an existing project, you must open the *Folder* containing the project. Once you have created a project, you can create new source files which must have the extension `lean`.

Tips for working with Lean

Tip: Infoview

Enter `ctrl-shift-enter` to open the Lean Infoview where hypotheses, goals and errors are displayed. To understand the effect of applying a tactic, I have found it helpful to place the cursor just before the source line, and then to alternate between `Home` and `End` while looking at the Infoview.

Tip: tactic

If you hover over a tactic its specification will be displayed.

¹Only tactics are used in the proofs and they are applied one at a time.

For example, the specification of the tactic `exact` is

`exact e` closes the main goal if its target type matches that of `e`.

Tip: theorem

You can display the statement of a theorem by hovering over its name.

Theorems are defined to be widely applicable, so initially you will not be able to understand the full statement of a theorem. In this tutorial only natural numbers and integers are used, so you can interpret the theorems narrowly, for example,

```
dvd_mul_right (a b : Nat) : a ∣ a * b
```

Preamble of a Lean program

The Lean community has developed an extremely large mathematical library called `Mathlib`. You will work with the definitions and theorems defined there, which must be imported as the first lines in your Lean source file.

```
import Mathlib.Tactic
```

imports the proof tactics.

You will need at least one other library that contains the definitions and theorems of the data type that you want to work on, for example,

```
import Mathlib.Data.Nat.Prime
```

The examples here are enclosed in

```
namespace gentle
```

to ensure that their names do not clash with the names of theorems in `MathLib`. In addition,

```
open Nat
```

will often be used to open the namespace of `Nat` so that we can write `factorial` instead of `Nat.factorial`.

Preamble does not appear in the examples here but can be found in the Lean source files.

Syntax

Lean uses two types of comments:

- Line comments start with `--` and continue to end of the line.
- Range comments start with `/-` and continue to `-/`.

Within `Mathlib` the convention is to use spaces around each operator:

$$a \leq b \wedge b \leq a \rightarrow a = b$$

and I will do so in this tutorial. You may ignore this convention in your proofs: `a ≤ b ∧ b ≤ a → a = b`.

There is also a convention for naming theorems: the name is written in lower case with underscores between the parts of the name. The name specifies the meaning of the theorem in a few words or abbreviations, for example, `min_le_right` is $\min a \, b \leq b$ and `min_le_left` is $\min a \, b \leq a$.

References

Mathematical logic

- M. Ben-Ari. *Mathematical Logic for Computer Science (Third Edition)*, Springer, 2012.
An introduction to mathematical logic including both syntactical and semantic proof methods, as well as sections on temporal logic and program verification.
- M. Huth and M. Ryan. *Logic in Computer Science: Modelling and Reasoning about Systems (Second Edition)*, Cambridge University Press, 2004.
An introduction to mathematical logic with emphasis on natural deduction, temporal logic and model checking.

Lean resources

- J. Avigad and P. Massot. *Mathematics in Lean*.
https://leanprover-community.github.io/mathematics_in_lean/
This is a comprehensive presentation of the use of Lean to prove mathematical theorems. The proofs in this document are based on examples given there.
- J. Avigad, L. de Moura, S. Kong and S. Ullrich. *Theorem Proving in Lean 4*.
https://lean-lang.org/theorem_proving_in_lean4/
A formal presentation of Lean.
- The website of the Lean Community contains links to important resources and to the Zulip chat where you can ask questions.
<https://leanprover-community.github.io/>

Learning materials on Lean

- The games on the Lean Game Server are a fun way to start learning Lean.
<https://adam.math.hhu.de/>
- D. J. Velleman. *How to Prove It with Lean*.
<https://djvelleman.github.io/HTPIwL/>
This work focuses on proof techniques in mathematics; to simplify learning it uses tactics developed by the author.
- H. Macbeth. *The Mechanics of Proof*.
<https://hrmacbeth.github.io/math2001/>
This work focuses on proof techniques in mathematics; to simplify learning it uses tactics developed by the author.

Acknowledgment

I am indebted to the members of the Lean Community for their patience and help as I took my first steps in Lean.

2 Commutativity of the greatest common denominator

A theorem is declared like a function in a programming language with the name of the theorem, its parameters and the statement of the theorem. The keyword `by` introduces a *tactic proof*, which we will be using exclusively in this tutorial.

```
1  /-
2    Commutativity of gcd
3  -/
4  theorem gcd_comm (m n : N) :
5    Nat.gcd m n = Nat.gcd n m := by
```

The parameters `m n` which are declared as natural numbers have a logical meaning as universally quantified variables.

$$\forall m, n \in N : (\text{gcd}(m, n) = \text{gcd}(n, m)).$$

The proof of the theorem uses the fact that the division operator is antisymmetric: if $m|n$ and $m|n$ then $m = n$. Applying this theorem results in two new goals.

```
6  apply Nat.dvd_antisymm
7    -- dvd_antisymm: (m | n ∧ n | m) → m = n,
8    --   where m = gcd m n, n = gcd n m
9    -- Two new goals: gcd m n | gcd n m, gcd n m | gcd m n
```

When you place the cursor before line 6, Lean Inforview will display the *tactic state*:

```
1 goal
m n : N
⊢ Nat.gcd m n = Nat.gcd n m
```

This gives the *context* that `m n` are natural numbers and the current *goal* following the turnstile symbol \vdash . After the theorem `Nat.dvd_antisymm` is applied, the tactic state changes to

```
2 goals
case a
m n : N
⊢ Nat.gcd m n | Nat.gcd n m
case a
m n : N
⊢ Nat.gcd n m | Nat.gcd m n
```

By the anti-symmetry of division, you can prove an equality by proving that the two sides divide each other, thus creating two goals.

Tactic: **apply**

If you have a theorem $P \rightarrow Q$ and the goal matches^a Q then `apply $\vdash P \rightarrow Q$` removes the goal Q and adds P as the new goal.

If you have a theorem Q and the goal matches Q then `apply Q` removes the goal and there are no more goals.

^aTechnically, if the goal *unifies* with Q , but we don't explain the concept here.

$\text{gcd}(m, n)$ will divide $\text{gcd}(n, m)$ only if it divides both n and m , so we again have two new subgoals. Then, by definition, $\text{gcd}(m, n)$ is a common divisor so it divides both the left parameter m and the right parameter n .

```
8  -- First goal
9  apply Nat.dvd_gcd
10     -- dvd_gcd: (k | m ∧ k | n) → k | gcd m n,
11     --   where k = gcd m n, m = n, n = m
12     -- New goals are gcd m n | n and gcd m n | m
13  apply Nat.gcd_dvd_right
14     -- gcd_dvd_right: gcd m n | n,
15     --   where m = m, n = n
16  apply Nat.gcd_dvd_left
17     -- gcd_dvd_left: gcd m n | m
18     --   where m = m, n = n
```

Repeat the same proof for the second subgoal $\text{gcd}(n, m) \mid \text{gcd}(m, n)$.

```
19 -- Second goal
20 apply Nat.dvd_gcd
21     -- dvd_gcd: k | m ∧ k | n → k | gcd m n,
22     --   where k = gcd n m, m = m, n = n
23     -- New goals are gcd n m | m and gcd n m | n
24  apply Nat.gcd_dvd_right
25     -- gcd_dvd_right: gcd m n | n,
26     --   where m = n, n = m
27  apply Nat.gcd_dvd_left
28     -- gcd_dvd_left: gcd m n | m
29     --   where m = n, n = m
30  done
```

Tip: **done**

All proofs should be terminated by `done`. This is not necessary but if your proof is not complete, `done` will display a message.

Tip: **Division**

The division operator in Lean is not the `|` symbol on your keyboard, but a similar Unicode symbol `|` obtained by typing `\|`.

3 Commutativity of minimum

To prove that the minimum operator is commutative, a hypothesis named `h` is introduced. Of course we have to prove the hypothesis before we can use it.

```
1  /-
2    Commutativity of minimum
3  -/
4  theorem min_comm (a b : N) :
5    min a b = min b a := by
6
7    have h : ∀ x y : N, min x y ≤ min y x := by
8      -- Hypothesis: For all natural numbers x y,
9      --   min x y ≤ min y x
```

Tactic: have

Introduces a new named hypothesis for use in the proof.

Tip: indentation

All statements used to prove a hypothesis must be indented a fixed number of spaces.

The hypothesis is $\forall m, n \in N : (\min(m, n) = \min(n, m))$. For the bound variables m, n , free variables must be substituted. The tactic `intro x y` introduces new free variables $x y$ for $m n$.

```
10  intro x y
11  -- Introduce arbitrary x y in place of ∀
```

Tactic: intro

Introduces free variables in place of bound variables in a universally quantified formula. A universally bound variable means that the formula has to hold for an arbitrary value so we substitute a variable that is this arbitrary value.

If the goal is $\vdash P \rightarrow Q$, `intro` introduces P as a hypothesis. If Q is proven then the hypothesis can be *discharged* and $P \rightarrow Q$ is proven.

The first step of the proof of the hypothesis uses the theorem that $c \leq a \wedge c \leq b \rightarrow c \leq \min(a, b)$. The Lean theorem `le_min` is applied to obtain two subgoals whose proofs complete the proof of the hypothesis `h`.

```
12  apply le_min
13    -- le_min: for all natural numbers c,
14    --   (c ≤ a ∧ c ≤ b) → c ≤ min a b
15    --   where a = y, b = x, c = min x y
16    --   New goals are min x y ≤ y and min x y ≤ x
```

```

17     apply min_le_right
18     -- min_le_right:  $\min a\ b \leq b$ 
19     --   where  $a = x$ ,  $b = y$ 
20     -- Solves goal  $\min x\ y \leq y$ 
21     apply min_le_left
22     -- min_le_left:  $\min a\ b \leq a$ 
23     --   where  $a = y$ ,  $b = x$ 
24     -- This completes the proof of h

```

Now that h has been proved, the indentation is removed to continue the proof the main theorem. `le_antisymm` is applied to the goal creating two subgoals which are proved by applying the hypothesis h .

```

25     apply le_antisymm
26     -- le_antisymm:  $(a \leq b \wedge b \leq a) \rightarrow a = b$ 
27     --   where  $a = \min a\ b$ ,  $b = \min b\ a$ 
28     --   New goals are  $\min a\ b \leq \min b\ a$  and  $\min b\ a \leq \min a\ b$ 
29     apply h
30     -- Apply h with  $x = a$  and  $y = b$ 
31     apply h
32     -- Apply h with  $x = b$  and  $y = a$ 
33     done

```

4 Cancellation properties

Here we prove theorems of arithmetic related to cancellation properties:

$$-a + (a + b) = b, \quad a + c = a + c \rightarrow b = c, \quad a * 0 = 0.$$

The library `import Mathlib.Data.Int.Basic` is imported because negative numbers are integers, not natural numbers.

Tip: associativity

The associativity of the operators `+` and `*` is defined to be *left*:

`a + b + c` is `(a + b) + c` and `a * b * c` is `(a * b) * c`.

Use `add_assoc` and `mul_assoc` to change the associativity of an expression.

In order to prove $-a + (a + b)$ we first need to proof that it equals $(-a + a) + b$ and then set $-a + a = 0$.

This proof uses the tactic `rw` that *rewrites* an expression with another expression that has already been proved to be equal to it. After two rewrites, applying the theorem `zero_add b` results in exactly the expression needed to proof the goal.

```
1 theorem neg_add_cancel_left (a b : Int) :  
2   -a + (a + b) = b := by  
3   rw [← add_assoc]  
4   -- add_assoc: a + b + c = a + (b + c)  
5   -- Addition is left associative: a + b + c = (a + b) + c  
6   -- New goal is -a + a + b = b  
7   rw [add_left_neg]  
8   -- add_left_neg: -a + a = 0  
9   -- New goal is 0 + b = b  
10  exact zero_add b  
11  -- zero_add: 0 + a = a, where a = b  
12  done
```

Tactic: `rw`

`rw [eqn]`, where `eqn` is an equation or an equivalence, rewrites the goal by replacing occurrences of the the left-hand side of `eqn` with the right-hand side.

`rw [← eqn]` rewrites the goal by replacing occurrences of the right-hand side of `eqn` with the left-hand side.

`rw [eqn] h` and `rw [← eqn] h` rewrite expressions in the hypothesis `h`.

The arrow `←` which is defined as “rewrites the goal by replacing occurrences of the right-hand side of `eqn` with the left-hand side” can seem non-intuitive: when `← add_assoc` is used to rewrite the goal $-a + (a + b) = b$ above, we are rewriting the *left-hand side* of the goal. However, the directions are referring to the sides of `eqn` (here, $a + b + c = a + (b + c)$) where occurrences of *its* right-hand side are $a + (b + c)$ are replaced by *its* left-hand side $a + b + c$.

Tactic: `exact`

The tactic `exact h` is used when `h` is exactly the statement of the current goal, so the goal is now proved. `exact` is similar to `apply` but limited because it can only be used if the hypothesis exactly matches the goal.

The parameter `h : a + b = a + c` in the theorem `add_left_cancel` is a hypothesis that is used in the proof. It must be provided when the theorem is applied (see below).

```
13 theorem add_left_cancel {a b c : Int}
14   (h : a + b = a + c) : b = c := by
15   rw [← neg_add_cancel_left a b]
16   -- neg_add_cancel (above): -a + (a + b) = b
17   -- New goal is -a + (a + b) = c
18   rw [h]
19   -- New goal is -a + (a + c) = c
20   exact neg_add_cancel_left a c
21   -- neg_add_cancel (above): -a + (a + b) = b, where b = c
22 done
```

The theorem `add_left_cancel` has four parameters: `a b c` of type `Int` and a hypothesis `h`. To use the theorem we can write

```
apply add_left_cancel i j k h
```

for some values `i j k` of type `Int` and a hypothesis `h`. When the theorem is used in line 36 below, `h` has been replaced by its definition in line 25:

```
exact add_left_cancel (a * 0 + a * 0 = a * 0 + 0)
```

but the integer parameters are not given. When the hypothesis is matched against `a + b = a + c` in the statement of the theorem, we have

```
a = a * 0   b = a * 0   c = 0.
```

Clearly, all three expressions are of type `Int` so Lean can infer their values and types when they are used:

```
exact add_left_cancel (a * 0) (a * 0) 0 h
```

Tip: `implicit`

Parameters declared with braces, such as `{a b c : Int}` state that `a b c` are *implicit* bound variables in the theorem whose actual names and types can be inferred when the theorem is used.

The proof of the following theorem is similar to the previous two. First the hypothesis `h` is proved using `rw` three times and it is then used as a hypothesis in the application of `add_left_cancel`.

```

23 theorem mul_zero {a : Int} :
24   a * 0 = 0 := by
25   have h : a * 0 + a * 0 = a * 0 + 0 := by
26     rw [← mul_add]
27     -- Distribute multiplication over addition (reversed)
28     -- mul_add: a * (b + c) = a * b + a * c,
29     --   where a = a, b = 0, c = 0
30     -- New goal is a * (0 + 0) = a * 0 + 0
31     rw [add_zero]
32     -- add_zero: a + 0 = 0
33     -- New goal is a * 0 = a * 0 + 0
34     rw [add_zero]
35     -- h is proved
36   exact add_left_cancel h
37   -- add_left_cancel (above): if a + b = a + c then b = c,
38   --   where a = a * 0, b = 0, c = 0
39   done

```

5 There exists an infinite number of prime numbers

Assume to the contrary that there are finitely many prime numbers p_1, p_2, \dots, p_n . Consider $q = (p_1 p_2 \cdots p_n) + 1$. For any $1 < m < q$, m does not divide q because the remainder is 1. Therefore, q is a prime since it is divisible only by 1 and q .

There are two ways to define primes in Lean:

- `Nat.prime_def_lt`: p is prime if and only if $(2 \leq p) \wedge \forall m < p (m|p \rightarrow m = 1)$.
- `Nat.Prime.eq_one_or_self_of_dvd`: p is prime only if $m|p \rightarrow (m = 1 \vee m = p)$.²

The proof above follows the definition `Nat.prime_def_lt`, but the proof using

`Nat.Prime.eq_one_or_self_of_dvd`

is easier, because we need only work the hypothesis $m|p$ and not with the quantifier $\forall m < p$.

The proof starts with the *definition* of p as the smallest prime factor of $n! + 1$. A major part of the proof is to show that such a p does exist.

```

1 theorem exists_infinite_primes (n : N) :
2   ∃ p, n ≤ p ∧ Nat.Prime p := by
3     -- For all natural numbers n,
4     --   there exists a natural number p,
5     --   such that n ≤ p and p is a prime
6   let p := minFac (n ! + 1)
7     -- if n ! + 1 ≠ 1, p is its smallest prime factor

```

Tactic: `let`

`let` introduces a definition whose scope is local.

Now we prove the hypothesis that $n! + 1 \neq 1$.

```

8   have f1 : n ! + 1 ≠ 1 := by
9     apply Nat.ne_of_gt <| succ_lt_succ <| factorial_pos _
10    -- factorial_pos: n ! > 0
11    -- succ_lt_succ: m < n → succ m < succ n
12    --   where m = 0, n = n !
13    --   succ(essor of) n is the formal definition of n + 1
14    -- ne_of_gt: b < a → b ≠ a
15    --   where b = 1 and a = n ! + 1
16    -- <| means that the formula on its right is
17    --   the input to the one on its left

```

Let us look at `apply Nat.ne_of_gt <| succ_lt_succ <| factorial_pos _` in detail.

- `factorial_pos _` is the theorem that the value of any factorial is positive, here, $0 < n!$.

²This definition is *not* if and only if because $p = 1$ satisfies the right-hand side but 1 is not a prime.

- `succ_lt_succ` uses the definition of $+1$ as the successor function. It takes $0 < n!$ from `factorial_pos _` and proves that $0 + 1 < n! + 1$.
- `Nat.ne_of_gt` is the simple theorem that if $1 < n + 1!$ then $1 \neq n + 1!$, which is the hypothesis `f1` that we want to prove.

Tip: right-to-left

The symbol `<|` means to compute the expression to its right and pass it to the left.

From the hypothesis `f1` it follows that $n! + 1$ has a smallest prime factor and therefore p exists.

```

18   have pp : Nat.Prime p := by
19     apply minFac_prime f1
20     -- minFac_prime: if n ≠ 1 then
21     --   the smallest prime factor of n prime,
22     --   where n = n ! + 1
23     -- f1 proves pp

```

To prove $n \leq p$, we will prove the equivalent formula $\neg(n \geq p)$ which itself is equivalent to $(n \geq p) \rightarrow \text{False}$. The method is to introduce $n \geq p$ as a new hypothesis and derive a contradiction.

```

24   have np : n ≤ p := by
25     apply le_of_not_ge
26     -- le_of_not_ge: ¬a ≥ b → a ≤ b
27     --   where a = n, b = p
28     -- New goal is ¬n > p
29     intro h
30     -- By definition of negation, n ≥ p implies False
31     -- Assume n ≥ p and make False the new goal
32     --   to prove np by contradiction

```

Tip: negation

A negation $\neg p$ is *defined* as $p \rightarrow \text{False}$. The tactic `intro` makes p into a hypothesis and `False` into the goal. When there are no more goals, $\neg p$ has been proved (by contradiction).

On page 6 we noted that the tactic `intro` is applicable to goals or hypotheses of the form $P \rightarrow Q$, as is done here with $p \rightarrow \text{False}$.

Two final hypotheses are needed: $p|n!$ and $p|1$.

In the proof of the first hypothesis, the theorem `dvd_factorial` takes two parameters: the fact that the minimum prime factor of any number (`_`) is positive and the hypothesis `h` that $n \geq p$.

```

33   have h1 : p | n ! := by
34     apply dvd_factorial (minFac_pos _) h
35     -- minFac_pos: 0 < minFac n,
36     --   where _ means that this holds for any n
37     -- dvd_factorial: (0 < m ∨ m ≤ n) → m | n !
38     --   where m = p
39     -- p is natural so 0 < m and
40     -- p ≤ n by assumption (intro) h
41     -- h1 is proved

```

To prove the second hypothesis we use the theorem `Nat.dvd_add_iff_right` whose main operator is if-and-only-if $a \leftrightarrow b$. This theorem can be used in two ways:

- Modus ponens (`mp`): $a \rightarrow b$, assume a and prove b .
- Modus ponens reversed (`mpr`): $a \leftarrow b$, assume b and prove a .

Here the reversed modus ponens is used.

```

42   have h2 : p | 1 := by
43     apply (Nat.dvd_add_iff_right h1).mpr (minFac_dvd _)
44     -- minFac_dvd: minFac n | n,
45     --   where _ means that this holds for any n
46     -- dvd_add_iff_right: k | m → (k | n ↔ (k | m) + n)
47     --   where k = p, m = n !, n = 1
48     -- p | n ! by h1, so p | 1 iff p | n ! + 1
49     -- mpr (MP reverse): p | n ! + 1 → p | 1
50     -- p | n ! + 1 by definition
51     -- since pp shows that p is prime
52     -- p | is proved

```

Applications of the hypotheses `pp` and `h2` complete the proof of `np`.

```

53   apply Nat.Prime.not_dvd_one pp
54   -- if p is a prime (true by pp) then ¬p | 1
55   --   which is p | 1 → False
56   exact h2
57   -- Use MP with h2, proving False and
58   --   thus np by contradiction,
59   --   since by definition a prime is ≥ 2
60   -- The proof of np : np: n ≤ p is (finally) complete

```

The current goal is now $\exists p(n \leq p \wedge \text{prime}(p))$. The existentially quantified variable p must be replaced with a specific value which is also p using the tactic `use`. This leaves the conjunction $n \leq p \wedge \text{prime}(p)$, which is split into two subgoals that are proved using the hypotheses `np` and `pp`.

```

61 use p
62   -- Introduce free variable p for the bound variable p
63   --   to get  $n \leq p \wedge \text{Nat.Prime } p$ 
64   -- Since both conjuncts are hypotheses,
65   --   the proof is complete
66 done

```

Tactic: use

Given an existential goal $\exists c \ A(c)$, this tactic introduces a free variable for the bound variable c to form a new goal. It will also attempt to prove the goal using the hypotheses.

Even without the comments the proof is not short. When proving complex theorems it is convenient to assume that certain lemmas are true, and then when the main proof is complete to return to prove the lemmas. In Lean the tactic `sorry` is accepted as a proof of anything. For example, the eight-line proof of the hypothesis `np` can be proved by `sorry`. Lean will issue a stern warning to tell you not to rest on your laurels.

```

theorem exists_infinite_primes_sorry (n : N) :
   $\exists p, n \leq p \wedge \text{Nat.Prime } p :=$  by
  let p := minFac (n ! + 1)
  have f1 :  $n ! + 1 \neq 1 :=$  by
    apply Nat.ne_of_gt <| succ_lt_succ <| factorial_pos _
  have pp : Nat.Prime p := by
    apply minFac_prime f1
  have np :  $n \leq p :=$  by sorry
  use p
done

```

Tactic: sorry

Proves any theorem.

6 The square root of two is irrational

Suppose that $\sqrt{2}$ is rational so that $\sqrt{2} = m/n$. Without loss of generality, assume that m, n are *coprime*, that is, they have no common factor. Then $m^2/n^2 = 2$ and $m^2 = 2n^2$, so 2 must divide m . Therefore, 2^2 must divide $2n^2$ and hence n is also divisible by 2, contradicting the assumption that m, n are coprime. This section contains a Lean proof of the central claim of the proof that if m, n are coprime then $m^2 \neq 2n^2$.

A lemma is just a different name for a theorem. This lemma proves that $a^2 = a \cdot a$ using `rw` on the successor function and the base case of the definition for taking the power of a number.

```

1 lemma pow_two (a : N) : a ^ 2 = a * a := by
2   rw [Nat.pow_succ]
3   -- Nat.pow_succ: n ^ succ m = n ^ m * n,
4   --   where n = a, m = 1, succ m = 1 + 1
5   -- New goal is a ^ 1 * a = a * a
6   rw [pow_one]
7   -- pow_one: a ^ 1 = a
8   done

```

The second lemma proves that if $2|m^2$ then $2|m$ using the first lemma and the theorem that if there is a prime factor of $a * b$ then it is a prime factor of either a or b . When the hypothesis h which is now $2|m \cdot m$ is rewritten, since $a = b = m$ the hypothesis becomes the disjunction $2|m \vee 2|m$.

```

9 lemma even_of_even_sqr (m : N)
10   (h : 2 | m ^ 2) : 2 | m := by
11   rw [pow_two] at h
12   -- pow_two (lemma): a ^ 2 = a * a,
13   --   where a = m
14   -- New goal is 2 | m
15   rw [prime_two.dvd_mul] at h
16   -- prime_two: 2 is prime
17   -- dvd_mul: if p is prime
18   --   then p | m * m ↔ p | m ∨ p | m,
19   --   where p = 2, m = m,
20   -- Apply to h : 2 | m * m → (2 | m ∨ 2 | m)
21   -- h is now 2 | m ∨ 2 | m, goal is still 2 | m

```

Split the disjunctive hypothesis $2|m \vee 2|m$ into two identical hypotheses; each one is exactly the goal of the lemma.

```

22 rcases h with h₁ | h₁
23   -- Splits disjunctive hypothesis h:
24   --   2 | m ∨ 2 | m into
25   --   two (identical) subformulas 2 | m, 2 | m

```

Prove both the (identical) subformulas.

```

26 · exact h1
27   -- 2 | m proves 2 | m
28 · exact h1
29   -- 2 | m proves 2 | m
30 done

```

Tactic: `rcases`

Given a hypothesis or goal that is a disjunction $A \vee B$ the tactic `rcases` splits it into two sub-hypotheses or subgoals A and B.

For a disjunctive hypothesis, prove the goal under both sub-hypotheses. For a disjunctive goal, prove one of the subgoals.

To prove the theorem, we assume $m^2 = 2n^2$ and try to prove a contradiction.

```

31 theorem sqr_not_even (m n : N) (coprime_mn : Coprime m n) :
32   m ^ 2 ≠ 2 * n ^ 2 := by
33   intro sqr_eq
34   -- Assume sqr_eq: m ^ 2 = 2 * n ^ 2 and
35   --   prove a contradiction

```

Add the hypothesis $2|m$. By the lemma `apply even_of_even_sqr`, it is sufficient to prove $2|m^2$ which becomes the new goal. Rewrite using the assumption $m^2 = 2n^2$ that was introduced to prove a contradiction and apply the trivial theorem that $a|a \cdot b$ proving the hypothesis.

```

36 have two_dvd_m : 2 | m := by
37   apply even_of_even_sqr
38   -- even_of_even_sqr (lemma): 2 | m ^ 2 → 2 | m
39   -- New goal is 2 | m ^ 2
40   rw [sqr_eq]
41   -- sqr_eq: m ^ 2 = 2 * n ^ 2.
42   -- Apply to the current goal.
43   -- The new goal is 2 | 2 * n ^ 2
44   apply dvd_mul_right
45   -- dvd_mul_right: a | a * b,
46   --   where a = 2, b = n ^ 2
47   -- Apply to the current goal to prove two_dvd_m : 2 | m

```

The definition of divisibility is: $a|b \leftrightarrow \exists c(a \cdot c = b)$. In Lean this is expressed by the forward direction (`mp`) of `dvd_iff_exists_eq_mul_left` applied to the hypothesis $2 | m$. The goal is now to find such a c .

```

48 have h : ∃ c, m = c * 2 := by
49   apply dvd_iff_exists_eq_mul_left.mp two_dvd_m
50   -- dvd_iff_exists_eq_mul_left: a | b ↔ ∃ c, b = c * a
51   --   where a = 2, b = m, c = c
52   -- Use MP with two_dvd_m: 2 | m to prove h

```

Given an existential formula such as $\exists c P(c)$, let c be some value that satisfies P .

```

53   rcases h with ⟨k, meq⟩
54   -- h :  $\exists c, m = c * 2$  is an existential formula
55   --   rcases on h:
56   --     k is the free variable for the bound variable c
57   --     meq :  $m = k * 2$  is a new hypothesis
58   -- Type < > using \< \>

```

Tactic: rcases

rcases h with ⟨v, h'⟩ means given a hypothesis h, let v be a value such that the new goal is h'.

By now you are certainly quite skillful in Lean, so the extent of the comments in the source code will be reduced.

We now prove a sequence of five hypotheses.

```

59   have h1 : 2 * (2 * k ^ 2) = 2 * n ^ 2 := by
60     rw [← sq_eq]
61     -- sq_eq :  $m^2 = 2 * n^2$ 
62     -- ← is right to left rewriting of  $2 * n^2$  in h1
63     -- New goal is  $2 * (2 * k^2) = m^2$ 
64     rw [meq]
65     -- Rewrite  $m = k * 2$  in h1
66     -- New goal is  $2 * (2 * k^2) = (k * 2)^2$ 
67     ring
68     -- Prove goal by using the ring axioms

```

Tactic: ring

Proves equalities that can be proved directly from the axioms of a commutative ring without taking any hypotheses into account. For the ring of integers, only addition, subtraction, multiplication and powers by natural numbers can be used. The division operation is not defined in the ring of integers because $1/2$ is not an integer.

It is easy to see that $2(2k^2) = (k \cdot 2)^2$ can be proved using only the definition of positive powers of integers as repeated multiplication, and the laws of associativity and commutativity of the integers.

The theorem `mul_right_inj'` is applied assuming that $2 \neq 0$, but this is a simple property of natural numbers which can be proved by the tactic `norm_num`.

```

69   have h2 : 2 * k ^ 2 = n ^ 2 := by
70     apply (mul_right_inj' (by norm_num : 2 ≠ 0)).mp h1
71     -- mul_right_inj':  $a \neq 0 \rightarrow (a * b = a * c \leftrightarrow b = c)$ 
72     --   where  $a = 2, b = 2 * k^2, c = n^2$ 
73     -- norm_num: solves equalities and inequalities like  $2 \neq 0$ 
74     -- Since  $2 \neq 0$ , MP on h1 proves h2

```

Tactic: norm_num

Proves numerical equalities and inequalities that do not use variables.

```

75   have h3 : 2 ∣ n := by
76     apply even_of_even_sqr
77     -- even_of_even_sqr (lemma) : 2 ∣ m ^ 2 → 2 ∣ m
78     -- where m = n
79     -- New goal is 2 ∣ n ^ 2
80     rw [← h2]
81     -- Rewrite right-to-left of h2 in the goal
82     -- New goal is 2 ∣ 2 * k ^ 2
83     apply dvd_mul_right
84     -- dvd_mul_right : a ∣ a * b,
85     -- where a = 2, b = k ^ 2 to prove h3

```

```

86   have h4 : 2 ∣ Nat.gcd m n := by
87     apply Nat.dvd_gcd
88     -- Nat.dvd_gcd : (k ∣ m ∧ k ∣ n) → k ∣ gcd m n
89     -- k = 2, m = m, n = n
90     -- New goals are 2 ∣ m and 2 ∣ n
91     · exact two_dvd_m
92     -- First goal is two_dvd_m
93     · exact h3
94     -- Second goal is h3

```

```

95   have h5 : 2 ∣ 1 := by
96     rw [Coprime.gcd_eq_one] at h4
97     -- if m and n are coprime then gcd m n = 1,
98     -- where m = 2 and n = 1
99     -- Apply to h4
100    -- New goals are 2 ∣ 1 and m, n are coprime
101    exact h4
102    -- Proves 2 ∣ 1
103    exact coprime_mn
104    -- Assumption that m, n are coprime

```

The sequence of hypotheses that have been proved terminates in $h_5 : 2 \mid 1$, but `norm_num` can prove that this is the negation of the true formula $1 \mid 2$, thereby deriving a contradiction.

```

105   norm_num at h5
106   -- Goal is 2 ∣ 1
107   -- norm_num can prove that this is False
108   -- Proving the contradiction of the initial assumption
109   done

```

7 Propositional logic

This section demonstrates tactics that can be used to prove theorems that use the operators of propositional logic. The final subsection presents the tactic for tautologies, which can immediately complete some proofs.

7.1 Conjunction and equivalence

When a theorem has equivalence (\Leftrightarrow) as the main operator and it must be split into two subgoals, one for modus ponens (mp) direction and one for the modus ponens reversed (mpr) direction.

```
1 theorem lt_iff_le_eq {a b : Int} :  
2   a < b  $\Leftrightarrow$  a  $\leq$  b  $\wedge$  a  $\neq$  b := by  
3   rw [lt_iff_le_not_le]  
4   -- lt_iff_le_not_le : a < b  $\Leftrightarrow$  (a  $\leq$  b  $\wedge$   $\neg$ b  $\leq$  a)  
5   constructor  
6   -- Create two subgoals (mp and mpr) from the current iff goal  
7   -- First subgoal is (a  $\leq$  b  $\wedge$   $\neg$ b  $\leq$  a)  $\rightarrow$  (a  $\leq$  b  $\wedge$  a  $\neq$  b)
```

Tactic: constructor

Splits a goal into two subgoals: equivalence (\Leftrightarrow) into two implications (Modus ponens and Modus ponens reversed) and conjunction (\wedge) into two conjuncts.

The current goal is an implication whose premise and conclusion are both conjunctions. We can use `intro` to introduce the premise as a hypothesis and then `rcases` (page 16) to split the conjunctive hypothesis into two subgoals to be proved. `intro h`

```
rcases h with <h0, h1>
```

It is possible to combine the two tactics into the tactic `rintro`.

```
8   -- Prove the mp goal  
9   · rintro <h0, h1>
```

We start with the implication of the mp direction, where `rintro` introduces the hypothesis and then `constructor` splits the conjunctive goal into two subgoals.

```
10  · rintro <h0, h1>  
11    -- rintro introduces the premise as a hypothesis and also  
12    -- performs an rcases on the hypothesis to split it  
13    -- two sub-hypotheses a  $\leq$  b and  $\neg$ b  $\leq$  a
```

```

14     constructor
15     -- Creates two subgoals from the current conjunctive goal
16     · exact h0
17     -- Proves the second subgoal
18     · intro h2
19     -- a ≠ b is a = b → False
20     apply h1
21     -- Replace False with the negation of the hypothesis
22     rw [h2]
23     -- Proof is complete since b ≤ b

```

The proof of the implication of the mpr direction is similar.

```

24     · rintro <h0, h1>
25     constructor
26     · exact h0
27     · intro h2
28     apply h1
29     apply le_antisymm h0 h2
30     -- le_antisymm: (a ≤ b) → (b ≤ a) → a = b
31 done

```

Tactic: **rintro**

Performs `intro` and then `rcases` to split the resulting hypothesis.

7.2 Disjunction

We now prove a theorem with an equivalence and a disjunction. In the previous theorem, we first split the equivalence into two implications. Here, tactic `rcases` is used with a theorem to split on the sign of y .

```

1 theorem lt_abs {x y : Int} :
2     x < |y| ↔ x < y ∨ x < -y := by
3     rcases le_or_gt 0 y with h | h
4     -- Absolute value depends on sign of y
5     -- le_or_gt 0 y: a ≤ b ∨ a > b

```

The equivalence goal is unchanged, but we are tasked with proving it under both hypotheses: $0 \leq y$ and $0 > y$. First we prove for $0 \leq y$, in which case we can rewrite $|y|$ by y .

```

6     · rw [abs_of_nonneg h]
7     -- abs_of_nonneg: 0 ≤ a → |a| = a
8     constructor
9     -- Split iff into mp and mpr
10    · intro h'

```

In the `mp` formula, the premise is introduced as a hypothesis and the conclusion becomes the goal $x < y \vee x < -y$. When a *goal* is a disjunction, it is sufficient to prove one disjunct. (Of course, if the *hypothesis* is a disjunction, we have to prove the theorem for *both* possibilities.) Here we are smart enough to tell Lean that we want to prove the left disjunct, because the right one won't make any progress toward the proof.

```

11     left
12     -- The current goal is a disjunction
13     --   so tell Lean which disjunct we want to prove
14     exact h'
15   . intro h'
16   rcases h' with h' | h'
17     -- The hypothesis is a disjunction and we have to prove
18     --   the goal for each disjunct
19   . exact h'
20   . linarith
21     -- The hypotheses are  $0 \leq y$  and  $x < -y$ 
22     -- Lean can prove that this implies the goal  $x < y$ 

```

Tactic: `left`, `right`

If the goal is a disjunction, tell Lean which disjunction you want to prove.

Tactic: `linarith`

The tactic solves linear equalities and inequalities. Unlike `ring` it can use hypotheses and unlike `norm_num` it can solve equations with variables.

The proof for $y < 0$ is similar.

```

23   . rw [abs_of_neg h]
24   constructor
25   . intro h'
26     right
27     exact h'
28   . intro h'
29     rcases h' with h' | h'
30     . linarith
31     . exact h'
32 done

```

7.3 Implication

The following theorem is proved by contradiction using the tactic `by_contra`, after which the proof is straightforward.

```

1 theorem T1a {A : Prop} : (¬A → A) → A := by
2   intro h1
3   by_contra h2
4     -- Prove A by contradiction: assume A and prove False
5   apply h2
6     -- Modus ponens
7   apply h1
8     -- Replace goal by the hypothesis
9   exact h2
10  done

```

Tactic: `by_contra`

This tactic removes a goal P , adds the hypothesis $\neg P$ and creates a new goal `False`.

Here is another proof of the same theorem, this time using contraposition instead of contradiction. The result of using the tactic `contrapose` will be unfamiliar. Given the *hypothesis* $\neg A \rightarrow A$, it does not change the hypothesis into $\neg A \rightarrow \neg\neg A$. Instead, the hypothesis becomes the goal $\neg(\neg A \rightarrow A)$ and the goal A becomes the hypothesis $\neg A$.

This makes sense by the deduction theorem.

$$\{H_1, H_2, \dots, H_n\} \vdash G$$

means

$$\vdash H_1 \wedge H_2 \wedge \dots \wedge H_n \rightarrow G$$

whose contrapositive is

$$\vdash \neg G \rightarrow \neg(H_1 \wedge H_2 \wedge \dots \wedge H_n),$$

which using deduction is

$$\neg G \vdash \neg(H_1 \wedge H_2 \wedge \dots \wedge H_n).$$

Although $\neg(\neg A \rightarrow A)$ doesn't simplify the proof, if we push negation inward (do it yourself!) the result is $\neg A \wedge \neg A$ which is trivial to prove. The exclamation point in the tactic means that following the tactic `contrapose`, the tactic `push_neg` is called to push negation inward.

```

11 theorem T1b {A : Prop} : (¬A → A) → A := by
12   intro h1
13   contrapose! h1
14     -- Replace h1 by its contrapositive
15     -- Push negation inward (!)
16   constructor
17     -- Split conjunction
18   · exact h1
19   · exact h1
20  done

```

Tactic: contrapose

Transforms a goal into its contrapositive. Applied to a hypothesis, it makes the negation of the goal into a hypothesis and the negation of the hypothesis into the goal. An exclamation point following `contrapose` calls `tactic push_neg` on the resulting contrapositive.

The interesting step in the following proof is

```
rcases h1 with <h2, _>
```

The hypothesis `h1` is $A \wedge \neg B$ while the goal is A . We use `rcases` to split the hypothesis into two, A and $\neg B$. Since only one sub-hypothesis is sufficient to prove the goal, we don't even bother to give the sub-hypothesis $\neg B$ a name.

```
21 theorem T2 {A B : Prop} : ((A → B) → A) → A := by
22   intro h1
23   contrapose! h1
24   -- Replace h1 by its contrapositive
25   -- and push negation inward
26   constructor
27   -- Split conjunction
28   · contrapose! h1
29   -- Goal is A → B, h1 is ¬A
30   -- Make ¬(A → B) the hypothesis and A the goal
31   -- Push negation inward (!)
32   rcases h1 with <h2, _>
33   -- Split the hypothesis
34   · exact h2
35   -- Only need to use left subformula
36   -- of the conjunction
37   · exact h1
38   done
```

Tip: Don't care

When a name or value is syntactically required but you don't care what its value is, you can use the underscore symbol (`_`) instead.

The next theorem we want to prove $(A \rightarrow B) \vee (B \rightarrow C)$ has a disjunction operator as its main operator. We prefer to carry out the proof using only the implication operator, so we first try to prove the hypothesis $\neg(A \rightarrow B) \rightarrow (B \rightarrow C)$.

When the hypothesis has been proved, use the `contrapose!` to negate the hypothesis and the goal and then exchange them. Moving the negations inward results in a hypothesis that is exactly the goal.

```
39 theorem T3 {A B C : Prop} : (A → B) ∨ (B → C) := by
40   have h1 : ¬(A → B) → (B → C) := by
41     -- Prove the implication equivalent to the disjunction
42     intro h2
43     intro h3
```

```

44     contrapose! h2
45     -- Contrapositive of  $\neg(A \rightarrow B)$ 
46     intro
47     -- No need to name the new hypothesis
48     exact h3
49     contrapose! h1
50     exact h1
51     done

```

7.4 Tautologies

Tautologies in propositional logic can be proved very easily using semantic methods such as truth tables and semantic tableaux. Lean can prove tautologies so the above theorems can be proved immediately using the tactic `tauto`.

```

theorem T1a {A : Prop} : ( $\neg A \rightarrow A$ )  $\rightarrow$  A := by tauto
theorem T2  {A B : Prop} : ((A  $\rightarrow$  B)  $\rightarrow$  A)  $\rightarrow$  A := by tauto
theorem T3  {A B C : Prop} : (A  $\rightarrow$  B)  $\vee$  (B  $\rightarrow$  C) := by tauto

```

Tactic: `tauto`

If a tautology can be formed from the hypotheses and the goal, the proof can be immediately completed using this tactic.

The following theorem that `tauto` works when the tautology is formed from both a hypothesis and a goal.

```
theorem T0a {A : Prop} (h :  $\neg A \rightarrow A$ ) : A := by tauto
```

This theorem shows that `tauto` works on substitution instances of a tautology.

```
theorem T0b {a : Nat} :  $\neg a = 0 \vee a = 0$  := by tauto
```

8 Induction

8.1 Sum of a sequence

The first program we prove is the one usually used to introduce induction.

$$\sum_{k=1}^n k = \frac{n(n+1)}{2}.$$

We import `Finset` for finite sets and `BigOperators` which enables the use of the operators \sum (`\sum`) and \prod (`\prod`).

```
1 import Mathlib.Tactic
2 import Mathlib.Data.Nat.Basic
3 import Mathlib.Algebra.BigOperators.Basic
4 namespace gentle
5 open Nat Finset BigOperators
```

Tip: `Finset`

Given a finite set of numbers `s` and a function `f` over these numbers, `Finset` defines the expressions $\sum x \text{ in } s \ f \ x$ and $\prod x \text{ in } s \ f \ x$.

Tip: `range`

`range n` is defined as the set of natural numbers less than `n`.

The linear display of expressions in Infoview is hard to follow so we will write the expressions in mathematical notation.

Since `range n` is $1, 2, \dots, n-1$, to get the set of natural numbers $0, 1, 2, \dots, n$ the proof will use `range (n + 1)`. We will use $r(n+1)$ for this sequence of numbers:

$$\sum_{i=r(n+1)} i = \frac{n(n+1)}{2}.$$

```
6 theorem sum_id (n : ℕ) :
7   Σ i in range (n + 1), i = n * (n + 1) / 2 := by
8   symm
9   have : 0 < 2 := by norm_num
```

This first step in the proof are to replace the goal with the symmetric goal

$$\frac{n(n+1)}{2} = \sum_{i=r(n+1)} i.$$

Tactic: `symm`

Replace an equality `t = u` by `u = t`. The tactic works for any symmetric relation.

The next step is to prove the trivial hypothesis $0 < 2$. The hypothesis is *anonymous* and is subsequently accessed as `this`.

```
10   have : 0 < 2 := by norm_num
```

Tip: Anonymous hypothesis

A hypothesis can be unnamed and can subsequently be accessed as `this`.

Division is replaced by multiplication which is easier to work with.

$$n(n+1) = 2 \sum_{i=r(n+1)} i.$$

```
11   apply Nat.div_eq_of_eq_mul_right this
12   -- div_eq_of_eq_mul_right:
13   --   (0 < n) ∧ (m = n * k) → m / n = k
```

The proof continues by induction, where `k` is the variable of induction and `ih` is the inductive hypothesis.

```
14   induction' n with k ih
```

Tactic: induction'

To prove a goal with `n` by induction, specify the induction variable `k` and a name `ih` for the inductive hypothesis.

The base case for $k = 0$ is proved using Lean's simplifier.

$$0(0+1) = 2 * \sum_{i=0(0+1)} i = 2 \cdot 0(0+1)$$

```
15   · simp
```

Tactic: simp

The `simp` tactic uses lemmas and hypotheses in `Mathlib` to simplify a goal or hypothesis.

For the inductive step we assume that the theorem holds for k and prove that it holds for $k+1$. The proof starts by separating out the last term of the sum. The notation is a bit clumsy because we aren't replacing $(k+1)+1$ by $k+2$.

$$(k+1)((k+1)+1) = 2 \sum_{i=r((k+1)+1)} i = 2 \left(\sum_{i=r(k+1)} i + (k+1) \right).$$

```
16   · rw [Finset.sum_range_succ]
17   -- Finset.sum_range_succ:
18   --   ∑ x in range (n + 1), f x = ∑ x in range n, f x + f n
```

Next, the distributive law is used.

$$(k+1)((k+1)+1) = 2 \sum_{i=r(n)} i + 2(k+1).$$

```
19   rw [mul_add 2]
20   -- mul_add: a * (b + c) = a * b + a * c
```

Substitute the inductive hypothesis.

$$(k+1)((k+1)+1) = k * (k+1) + 2(k+1).$$

```
21   rw [← ih]
```

Lean is using the successor function which is now changed to +1 and the proof concludes with a simple calculation that can be performed by the tactic `ring`.

```
22   rw [succ_eq_add_one]
23   -- succ_eq_add_one : succ n = n + 1
24   ring
25   done
```

8.2 Inductive definition of addition

This subsection demonstrates inductive definitions and the tactic `rf1` (*reflexive*).

Here is an inductive definition of natural numbers. The program is enclosed in the namespace `gentle` to avoid clashing with predefined `Nat`. After defining `Nat` we open its namespace so that we can use `zero` and `succ` without the prefix `Nat`.

```
1   namespace gentle
2
3   inductive Nat
4   | zero : Nat
5   | succ : Nat → Nat
6
7   namespace Nat
```

The next step is to defined addition inductively.

```
8   def add : Nat → Nat → Nat
9   | x, zero => x
10  | x, succ y => succ (add x y)
```

Although we normally think of addition as a function of two variables $add(a, b)$, there are advantages to defining it as a function that takes one variable $add(a)$ resulting in another function $add_a(b)$ that gives the sum of a and b .³

³This is called *currying* but an explanation is outside the scope of this tutorial.

The first theorem we proof is $0 + n = n$ for all natural numbers n . The proof is by induction and the base case $0 + 0 = 0$ follows by `rfl` from the base case of the definition of `add`.

```

11 theorem zero_add (n : Nat) :
12     add zero n = n := by
13     induction' n with k ih
14     · rfl

```

Tactic: `rfl`

`rfl` solves a goal that is an equality if they are *definitionally equivalent*, that is, they define the same entity.

Consider the goal:

$$(m + k) + 1 + 1 = (m + (k + 1)) + 1$$

By definition of `add`, $(a + (b + 1))$ is $(a + b) + 1$ so the right-hand side of the equation $(m + (k + 1)) + 1$ means $((m + k) + 1) + 1$ which is the left-hand side. Therefore, the tactic `rfl` solves it. You should similarly check the other uses of `rfl` below.

The inductive step is to prove $0 + (k + 1) = (0 + k) + 1$. Using definition of `add` to rewrite the left-hand side, we get $(0 + k) + 1 = k + 1$ which can be solved using the inductive hypothesis that $0 + k = k$.

```

15     · rw [add]
16       rw [ih]
17     done

```

Next we prove $(m + 1) + n = (m + n) + 1$ by induction. The base case $(m + 1) + 0 = m + 1$ is immediate from the base case of the definition of `add`.

```

18 theorem succ_add (m n : Nat) :
19     add (succ m) n = succ (add m n) := by
20     induction' n with k ih
21     · rfl

```

The inductive step is to prove $(m + 1) + (k + 1) = (m + (k + 1)) + 1$. From the definition of `add` it follows that $((m + 1) + k) + 1 = (m + (k + 1)) + 1$.

```

22     · rw [add]

```

Substituting the inductive hypothesis $(m + 1) + k = (m + k) + 1$ results in $((m + k) + 1) + 1 = (m + (k + 1)) + 1$, which, as noted above, is definitionally equivalent and can be solved by `rfl`.

```

23         rw [ih]
24         rfl
25     done

```

8.3 Multiple steps in a proof

In this section we explain two ways to shorten proofs in Lean by performing more than one step in the same action. We start with the proof of $(m|n \vee m|k) \rightarrow m|mk$ using methods already encountered.

Since the hypothesis is a disjunction, the goal must be proved separately for each disjunct using `rcases` to create separate hypotheses. Furthermore, since $m|n$ is defined as $\exists a(n = ma)$ so new variables must be introduced (page 16).

```
1 theorem div_prod1 {m n k : ℕ} (h : m | n ∨ m | k) :  
2   m | n * k := by  
3   rcases h with ⟨a, h1⟩ | ⟨b, h2⟩
```

The new hypotheses are `h1 : n = m * a` and `h2 : k = m * b` which can be used to rewrite the goals. The rest of the proof is straightforward.

```
4   · rw [h1]  
5     rw [mul_assoc]  
6     apply dvd_mul_right  
7     -- dvd_mul_right: a | a * b  
8   · rw [h2]  
9     rw [mul_comm]  
10    rw [mul_assoc]  
11    apply dvd_mul_right  
12  done
```

Proofs can be shortened by sequentially rewrites of the goal.

```
· rw [h1, mul_assoc]  
· rw [h2, mul_comm, mul_assoc]
```

The hypotheses `h1` and `h2` replace `m | n` and `m | k` by the definitionally equivalent `n = m * a` and `k = m * b`, so `rfl` can be used to perform the substitutions without creating hypotheses.

```
1 theorem div_prod2 {m n k : ℕ} (h : m | n ∨ m | k) :  
2   m | n * k := by  
3   rcases h with ⟨a, rfl⟩ | ⟨b, rfl⟩  
4   · rw [mul_assoc]  
5     apply dvd_mul_right  
6   · rw [mul_comm, mul_assoc]  
7     apply dvd_mul_right  
8  done
```

A Keyboard shortcuts

Lean uses Unicode symbols that are not on keyboards. This table shows keyboard shortcuts that are used in VSC for entering the symbols. Enter a space or tab after a shortcut and the symbol will appear. If you hover over an symbol the shortcut will be displayed.

Unicode symbol	Keyboard shortcut
	\
\mathbb{N}	\N
\leq	\le, \leq
\geq	\ge, \geq
\neq	\ne, \neq
\leftarrow	\l
\rightarrow	\r, \imp
\Leftrightarrow	\iff
\wedge	\and
\vee	\or
\forall	\all, \forall
\exists	\ex, \exists
\neg	\n, \neg
$<$	\<
$>$	\>
\vdash	\ -, \vdash
h_1	h\1
Σ	\sum
\prod	\prod

B Tactics

Tactic	Page	Definition
<code>apply</code>	4	If you have a theorem $P \rightarrow Q$ and the goal matches Q then <code>apply $\vdash P \rightarrow Q$</code> removes the goal Q and adds P as the new goal. If you have a theorem Q and the goal matches Q then <code>apply Q</code> removes the goal and there are no more goals.
<code>by_contra</code>	22	This tactic removes a goal P , adds the hypothesis $\neg P$ and creates a new goal <code>False</code> .
<code>contrapose</code>	23	Transforms a goal into its contrapositive. Applied to a hypothesis, it makes the negation of the goal into a hypothesis and the negation of the hypothesis into the goal. An exclamation point following <code>contrapose</code> calls tactic <code>push_neg</code> on the resulting contrapositive.
<code>constructor</code>	19	Splits a goal into two subgoals: equivalence (\leftrightarrow) into two implications (Modus ponens and Modus ponens reversed) and conjunction (\wedge) into two conjuncts.
<code>exact</code>	9	The tactic <code>exact h</code> is used when h is exactly the statement of the current goal, so the goal is now proved. <code>exact</code> is similar to <code>apply</code> but limited because it can only be used if the hypothesis exactly matches the goal.
<code>have</code>	6	Introduces a new named hypothesis for use in the proof.
<code>induction'</code>	26	To prove a goal with n by induction, specify the induction variable k and a name ih for the inductive hypothesis.
<code>intro</code>	6	Introduces free variables in place of bound variables in a universally quantified formula. A universally bound variable means that the formula has to hold for an arbitrary value so we simply substitute a variable that is this arbitrary value. If the goal is $\vdash P \rightarrow Q$ then <code>intro</code> introduces P as a hypothesis. If Q is proven then the hypothesis can be <i>discharged</i> and $P \rightarrow Q$ is proven. For an existential goal $\exists a A(a)$, <code>intro c</code> introduces $A(c)$ as the new goal.
<code>left, right</code>	21	If the goal is a disjunction, tell Lean which disjunction you want to prove.
<code>linarith</code>	21	The tactic solves linear equalities and inequalities. Unlike <code>ring</code> it can use hypotheses and unlike <code>norm_num</code> it can solve equations with variables.
<code>norm_num</code>	18	Proves numerical equalities and inequalities that do not use variables.
<code>range</code>	25	<code>range n</code> is defined as the set of natural numbers less than n .
<code>rcases</code>	16	Given a hypothesis or goal that is a disjunction $A \vee B$ the tactic <code>rcases</code> splits it into two sub-hypotheses or subgoals A and B . For a disjunctive hypothesis, prove the goal under both sub-hypotheses. For a disjunctive goal, prove one of the subgoals.
<code>rcases</code>	17	<code>rcases h with $\langle v, h' \rangle$</code> means given a hypothesis h , let v be a value such that the new goal is h' .
<code>rfl</code>	28	<code>rfl</code> solves a goal that is an equality if they are <i>definitionally equivalent</i> , that is, they define the same entity.

ring	17	Proves equalities that can be proved directly from the axioms of a commutative ring without taking any hypotheses into account. For the ring of integers, addition, subtraction, multiplication and powers by natural numbers, but not division, can be used.
rintro	20	Performs <code>intro</code> and then <code>rcases</code> to split the resulting hypothesis.
rw	8	<code>rw [eqn]</code> , where <code>eqn</code> is an equation or an equivalence, rewrites the goal by replacing occurrences of the left-hand side of <code>eqn</code> with the right-hand side. <code>rw [← eqn]</code> rewrites the goal by replacing occurrences of the right-hand side of <code>eqn</code> with the left-hand side. <code>rw [eqn] h</code> and <code>rw [← eqn] h</code> rewrite expressions in the hypothesis <code>h</code> .
simp	26	The <code>simp</code> tactic uses lemmas and hypotheses in <code>Mathlib</code> to simplify a goal or hypothesis.
sorry	14	Proves any theorem.
symm	25	Replace an equality $t = u$ by $u = t$. The tactic works for any symmetric relation.
use	14	Given an existential goal $\exists c \ A(c)$, this tactic introduces a free variable for the bound variable <code>c</code> to form a new goal. It will also attempt to prove the goal using the hypotheses.

C Tips

Topic	Page	Tip
Anonymous hypothesis	26	A hypothesis can be unnamed and can be subsequently accessed as <code>this</code> .
Associativity	8	The default associativity in Lean is <i>left</i> : $a + b + c$ means $(a + b) + c$. Even if the operation is associative, you still have to prove associativity for a given expression.
Division	5	The division operator in Lean is not the <code> </code> symbol on your keyboard, but a similar Unicode symbol <code> </code> obtained by typing <code>\ </code> .
done	5	All proofs should be terminated by <code>done</code> . This is not necessary but if your proof is not complete, <code>done</code> will display a message.
Don't care	23	When a name or value is syntactically required but you don't care what its value is, you can use the underscore symbol (<code>_</code>) instead.
Finset	25	Given a finite set of numbers s and a function f over these numbers, <code>Finset</code> defines the expressions $\sum x \text{ in } s f x$ and $\prod x \text{ in } s f x$.
Indentation	6	All statements used to prove a hypothesis must be indented a fixed number of spaces.
Infoview	1	Enter <code>ctrl-shift-enter</code> to open the Lean Infoview where hypotheses, goals and errors are displayed. I have found it helpful to place the cursor just before the source line, and then to alternate between <code>Home</code> and <code>End</code> while looking at the Infoview.
Implicit	9	Parameters declared with braces, such as <code>{a b c : Int}</code> state that $a b c$ are <i>implicit</i> bound variables in the theorem whose actual names and types can be inferred when the theorem is used.
Negation	12	A negation $\neg p$ is defined as $p \rightarrow \text{False}$ so it is proved by introducing p as a hypothesis and then showing that this results in a contradiction by proving <code>False</code> .
range	25	<code>range n</code> is defined as the set of natural numbers less than n .
Right-to-left	12	The symbol <code>< </code> means to compute the expression to its right and pass it to the left.
Tactic	1	If you hover over a tactic its specification will be displayed.
Theorem	2	You can display the statement of a theorem by hovering over its name.